


Application Note		資料作成 100210	資料番号 an2k-017
テーマ	1mSec で AD サンプリングし CUnet で EXCEL に入力する		
使用機器	MPC-2000 システム、Windows XP、MS-EXCEL 他		

■機器構成



■概要

- ・ 定速でパルスモータを回しながら 1mSec 単位で AD を読みます。
- ・ 計測中はデータを MBK 配列に格納し、計測終了後 CUnet メール転送でワークシートにデータを入力、グラフを描画します。
- ・ このサンプルは MPC-2000 で作成しましたが、MPC-1000/2100 でも可能です。
- ・ MPC-AD12 の AD レンジは標準(搭載デバイス AD7890-4)で 0~4095mV です。
- ・ EXCEL は 2003 を使用しました。

■MPC プログラムの注意点

1mSec で読み込むためには効率良くタスクを回さなければなりません。このプログラムは実機を想定してダミータスクを走らせていますが、その中に TIME や SWAP を入れてタスクディスパッチの効率を上げています。

例えば、*DUMMY の※1 の SWAP がある場合、プログラムを Ctrl+A で停止したとき

```
# *0_ [100] *1 [210] *10 [530] *11 [530]
*12 [530] *13 [490] *14 [470] *15 [530]
*16 [530] *17 [530] *18 [530] *19 [530]
*20 [530]
```

#

のように表示されますが、SWAP を無くしてしまうと

```
# *0_ [100] *1 [660] *10! [520] *11! [520]
*12! [520] *13! [520] *14! [520] *15! [520]
*16! [520] *17! [520] *18! [520] *19! [520]
*20! [520]
```

!は時間浪費タスクです。

#

というように!マークが付くタスクが発生します。これらは効率の悪いタスクで、これがあると 1mSec の入力 はできなくなります。

TIME や SW(), IN() には SWAP と同等の機能が含まれますが、FOR~NEXT や IF で演算が連続したり、空っぽの DO ~LOOP 等は要注意です。

マルチタスクで高速計測をする場合、計測している間は他のタスクを PAUSE させるのも有効な手段です。

■MPC プログラム

青文字が AD の主要な部分です。

```
CUNET 4 4 7          /* SA, OWN, EN
TIME 5
CU_POST             /* CUnet サーバ起動

QUIT 1
FORK 1 *PULSE_AD    /* PULSE, AD

FOR I=10 TO 20      /* 実機を想定したダミータスク
  QUIT I
  FORK I *DUMMY
NEXT I

END

*PULSE_AD
PG 0
GOSUB *XY03_HOME_X /* 原点復帰
DO
  ACCEL X_A 20000 1000 1000 /* 0 位置に戻る時のスピード
  MOVS X_A 0
  WAIT RR(X_A)==0
  ACCEL 10000 1000 10000 /* 計測時のスピード MAX=MIN(加減速無しの定速移動)

  PRINT "Wait Start from PC"
  I=1000
  WAIT SW(2000)==1 /* PC→MPC スタート
  FILL MBK(1000) 2000 0 /* 使用エリアクリア

  RMVS X_A 10000 /* X 軸パルス発生
  SYSCLK=1000
  DO
    WAIT I<=SYSCLK /* 1mSec 待ち。これがないと 8000 を超える
    SC=SYSCLK /* 確認用
    MBK(1)=AD(0) /* AD 入力
    IF RR(X_A)==0 THEN /* X 軸パルス停止待ち
      BREAK
    END_IF
    I=I+1
  LOOP

  PR I SC
  IF I<>SC THEN /* サンプリング回数と時間の確認
    PRINT "TIME ERROR"
  END_IF
  OUT I-1000 2036~Lng /* データ数

  PRINT "DATA SET COMPLETED"
  ON 2256 /* MPC→PC 計測終了
  WAIT SW(2000)==0
  OFF 2256

LOOP

*DUMMY
DO
  ON 0 1 2
  TIME 100
  OFF 0 1 2
  TIME 100
  A_=0
  FOR A_=0 TO 1000
    B_=A_*A_
    SWAP /* ※1
  NEXT A_
  WAIT SW(192)==0
LOOP

*XY03_HOME_X /* X 軸原点復帰
ACCEL X_A 10000 100 100 /* 原点復帰スピード
IF HPT(X_IN0)<>0 THEN /* X 軸 IN0 がオンなら退避移動
  RMVS X_A 10000
```

```

    WAIT RR(ALL_A)==0
END_IF
TIME 100
SHOM X_A INO_ON
HOME -100000 0 0 0
WAIT RR(ALL_A)==0
STPS 0 0 VOID VOID      /* X,Y軸ここを'0'にセット
PRINT "HOME"
RETURN

```

■実行後の MPC の MBK データ

第1パラメータはAD値、第2パラメータはMBK配列番号

```

s_mbk 1626 1000
s_mbk 1626 1001
s_mbk 1621 1002
s_mbk 1621 1003
s_mbk 1584 1004
s_mbk 1584 1005
s_mbk 1563 1006
(以下略)

```

■VBA Module1.bas コード (使っている関数のみ)

```

' CUnet 初期化、情報
Declare Function cunet_usb_open Lib "usbcunet.dll" () As Long
Declare Function cunet_dll_ver Lib "usbcunet.dll" () As Long
Declare Function cunet_fw_ver Lib "usbcunet.dll" () As Long
Declare Sub cunet_init Lib "usbcunet.dll" (ByVal sa As Long, ByVal ow As Long, ByVal en As Long)

' CUnet グローバルメモリ
Declare Function cunet_in Lib "usbcunet.dll" _
    (ByVal adr As Long, ByVal siz As Long) As Long
Declare Sub cunet_on Lib "usbcunet.dll" (ByVal adr As Long)
Declare Sub cunet_off Lib "usbcunet.dll" (ByVal adr As Long)
Declare Function cunet_sw Lib "usbcunet.dll" (ByVal adr As Long) As Long

' CUnet メール転送
Declare Function cunet_req_mbk Lib "usbcunet.dll" _
    (ByVal req_sa As Long, ByVal ar_top As Long, ByRef rcv_ar As Any) As Long

' cunet Register
Declare Function cunet_chk_mfr Lib "usbcunet.dll" (ByVal sa As Long) As Long

' cunet 定数
Public Const Cu_Lng As Long = 8 ' 4Byte 符号付

Public toral_rc As Long

Public Declare Sub Sleep Lib "kernel32" (ByVal dwMilliseconds As Long)
Public Declare Function timeGetTime Lib "winmm.dll" () As Long

```

■VBA UserForm1.frm コード

青文字がCUnetの関数です。
1000データを入力するのは約1.9秒でした(PCの性能により異なります)。

```

' Input ボタン
Private Sub CommandButton1_Click()
Dim ar(60) As Long ' 取得したデータを格納する配列

    CUSA = 4 ' MPC SA
    CommandButton1.Enabled = False
    CommandButton2.Enabled = False

    cunet_off 2000

Do
    Range("A1", "A10000").Select
    Selection.ClearContents
    Range("A1").Select
    Cells(1, 1) = "AD"

    While cunet_chk_mfr(CUSA) = 0 ' MPC SAの確認

```

```

    MsgBox ("missing SA" + CStr(CUSA))
Wend

cunet_on 2000                                ' MPC へ Start をかける
Label2.Caption = ""
Do Until cunet_sw(2256) = 1                    ' MPC のデータ取り込み終了待ち
    DoEvents
Loop
st1 = timeGetTime()

dtcnt = cunet_in(2036, Cu_Lng)                 ' データ数=CUnet グローバルメモリに書いてある
Label2.Caption = "Data Count=" + CStr(dtcnt)
rc = 0                                         ' ワークシート行カウント
toral_rc = 0                                  ' Graph 表示のための public 変数
brk = 0
retry = 0
Label3.Caption = ""
Label4.Caption = ""

For mbk = 1000 To dtcnt + 1000 Step 60        ' 60 点毎
    Do
        ' st2 = timeGetTime()
        res = cunet_req_mbk(CUSA, mbk, ar(0)) ' MBK データ取得 SA,top MBK(), input array
        ' rt2 = timeGetTime() - st2          ' 約 100msec/1 回受信
        If res = 0 Then Exit Do              ' 0 なら正常
        retry = retry + 1
    DoEvents
Loop

For i = 0 To 59
    Cells(rc + 2, 1).Value = ar(i)
    Label3.Caption = rc
    rc = rc + 1
    If rc >= dtcnt Then
        brk = 1
        Exit For
    End If
Next i
If brk = 1 Then Exit For
Next mbk
toral_rc = rc
Label3.Caption = "Retry " + CStr(retry)

cunet_off 2000
Do Until cunet_sw(2256) = 0
    DoEvents
Loop
rt1 = timeGetTime() - st1                      ' 約 1.9sec/1000 データ入力
Label4.Caption = "Time=" + CStr(rt1)

graph_draw

st1 = timeGetTime()                            ' LOOP の時グラフを見るためのディレイ
Do Until timeGetTime() > st1 + 1000
    DoEvents
Loop

If retry <> 0 Then CheckBox1.Value = False      ' LOOP させるか
If CheckBox1.Value = False Then Exit Do

Loop

CommandButton1.Enabled = True
CommandButton2.Enabled = True
CommandButton2.SetFocus

End Sub

' Graph ボタン
Private Sub CommandButton2_Click()

    graph_draw
    CommandButton1.SetFocus

End Sub

```

```

Private Sub Label3_Click()

End Sub

Private Sub UserForm_Initialize()
    Label2.Caption = ""
    CommandButton2.Enabled = False

    If cunet_usb_open() <> 1 Then
        MsgBox "USB CUnet Open Error"
    End
End If
Label1.Caption = "DLL Ver:" + CStr(cunet_dll_ver()) + " FW Ver:" + CStr(cunet_fw_ver())

cunet_init 255, 0, 0      ' USB-CUnet 初期化
Sleep 500
cunet_init 0, 4, 7

Label3.Caption = ""
Label4.Caption = ""

End Sub

Sub graph_draw()
    Dim WS As Worksheet

    graph_clear

    For Each WS In ActiveWindow.SelectedSheets
        sn = WS.Name
    Next
    toral_rc = toral_rc + 1
    Range("A1:A" + CStr(toral_rc)).Select
    Charts.Add
    ActiveChart.ChartType = xlLine
    ActiveChart.SetSourceData Source:=Sheets(sn).Range("A1:A" + CStr(toral_rc))
    ActiveChart.Location Where:=xlLocationAsObject, Name:=sn

End Sub

' 既存グラフを消去する
Function graph_clear() As Boolean
    On Error GoTo errorHandler
    Range("A1").Select

    nChart = ActiveSheet.ChartObjects.Count
    If nChart > 0 Then
        For cnt = 1 To nChart
            ActiveSheet.ChartObjects(1).Activate      ' 消すと count は減算されるので常に 1
            ActiveChart.ChartArea.Select
            ActiveWindow.Visible = False
            ActiveChart.Parent.Delete
            ' Selection.Delete
        Next
    End If

    Range("A1").Select

    graph_clear = True

    Exit Function

errorHandler:
    MsgBox "グラフ消去でエラーがありました (Err " + Str(Err) + ")"
    ' Resume Next
    ' Error Err
    graph_clear = False
End Function

Private Sub UserForm_QueryClose(Cancel As Integer, CloseMode As Integer)
    If CommandButton1.Enabled = False Then
        Cancel = True
    End If
End Sub

```

■実行中の EXCEL スクリーンショット

